Title

stata.com

asclogit — Alternative-specific conditional logit (McFadden's choice) model

Description

`asclogit` fits McFadden's choice model, which is a specific case of the more general conditional logistic regression model fit by `clogit`. The command requires multiple observations for each case (individual or decision), where each observation represents an alternative that may be chosen. `asclogit` allows two types of independent variables: alternative-specific variables, which vary across both cases and alternatives, and case-specific variables, which vary across only cases.

Quick start

McFadden's choice model of `y` from alternatives `alts` as a function of `x1` for cases identified by `idvar`

```
asclogit y x1, case(idvar) alternatives(alts)
```

As above, and add `x2` that varies across cases only

```
asclogit y x1, case(idvar) alternatives(alts) casevars(x2)
```

As above, but omit alternative-specific intercepts

```
asclogit y x1, case(idvar) alternatives(alts) casevars(x2) noconstant
```

Multinomial logit model if all covariates are case-specific

```
asclogit y, case(idvar) alternatives(alts) casevars(x1 x2)
```

Menu

Statistics > Categorical outcomes > Alternative-specific conditional logit

Syntax

asclogit *depvar* [*indepvars*] [*if*] [*in*] [*weight*] , case(*varname*)

alternatives(*varname*) [*options*]

depvar equal to 1 identifies the outcome or chosen alternatives, whereas a 0 indicates the alternatives that were not chosen. There can be multiple alternatives chosen for each case.

options	Description
Model	
* case(*varname*)	use *varname* to identify cases
* alternatives(*varname*)	use *varname* to identify the alternatives available for each case
casevars(*varlist*)	case-specific variables
basealternative(# | *lbl* | *str*)	alternative to normalize location
noconstant	suppress alternative-specific constant terms
altwise	use alternativewise deletion instead of casewise deletion
offset(*varname*)	include *varname* in model with coefficient constrained to 1
constraints(*constraints*)	apply specified linear constraints
collinear	keep collinear variables
SE/Robust	
vce(*vcetype*)	*vcetype* may be oim, robust, cluster *clustvar*, bootstrap, or jackknife
Reporting	
level(#)	set confidence level; default is level(95)
or	report odds ratios
noheader	do not display the header on the coefficient table
nocnsreport	do not display constraints
display_options	control columns and column formats, row spacing, line width, display of omitted variables and base and empty cells, and factor-variable labeling
Maximization	
maximize_options	control the maximization process; seldom used
coeflegend	display legend instead of statistics

* case(*varname*) and alternatives(*varname*) are required.

indepvars and *varlist* may contain factor variables; see [U] **11.4.3 Factor variables**.

bootstrap, by, fp, jackknife, and statsby are allowed; see [U] **11.1.10 Prefix commands**.

Weights are not allowed with the bootstrap prefix; see [R] **bootstrap**.

fweights, iweights, and pweights are allowed (see [U] **11.1.6 weight**), but they are interpreted to apply to cases as a whole, not to individual observations. See *Use of weights* in [R] **clogit**.

coeflegend does not appear in the dialog box.

See [U] **20 Estimation and postestimation commands** for more capabilities of estimation commands.

Options

Model

case(*varname*) specifies the numeric variable that identifies each case. case() is required and must be integer valued.

alternatives(*varname*) specifies the variable that identifies the alternatives for each case. The number of alternatives can vary with each case; the maximum number of alternatives cannot exceed the limits of tabulate oneway; see [R] **tabulate oneway**. alternatives() is required and may be a numeric or a string variable.

casevars(*varlist*) specifies the case-specific numeric variables. These are variables that are constant for each case. If there are a maximum of J alternatives, there will be $J-1$ sets of coefficients associated with the casevars().

basealternative(# | *lbl* | *str*) specifies the alternative that normalizes the latent-variable location (the level of utility). The base alternative may be specified as a number, label, or string depending on the storage type of the variable indicating alternatives. The default is the alternative with the highest frequency.

If vce(bootstrap) or vce(jackknife) is specified, you must specify the base alternative. This is to ensure that the same model is fit with each call to asclogit.

noconstant suppresses the $J-1$ alternative-specific constant terms.

altwise specifies that alternativewise deletion be used when marking out observations due to missing values in your variables. The default is to use casewise deletion; that is, the entire group of observations making up a case is deleted if any missing values are encountered. This option does not apply to observations that are marked out by the if or in qualifier or the by prefix.

offset(*varname*), constraints(*numlist* | *matname*), collinear; see [R] **estimation options**.

SE/Robust

vce(*vcetype*) specifies the type of standard error reported, which includes types that are derived from asymptotic theory (oim), that are robust to some kinds of misspecification (robust), that allow for intragroup correlation (cluster *clustvar*), and that use bootstrap or jackknife methods (bootstrap, jackknife); see [R] ***vce_option***.

Reporting

level(#); see [R] **estimation options**.

or reports the estimated coefficients transformed to odds ratios, that is, e^b rather than b. Standard errors and confidence intervals are similarly transformed. This option affects how results are displayed, not how they are estimated. or may be specified at estimation or when replaying previously estimated results.

noheader prevents the coefficient table header from being displayed.

nocnsreport; see [R] **estimation options**.

display_options: noci, nopvalues, noomitted, vsquish, noemptycells, baselevels, allbaselevels, nofvlabel, fvwrap(#), fvwrapon(*style*), cformat(%*fmt*), pformat(%*fmt*), sformat(%*fmt*), and nolstretch; see [R] **estimation options**.

Maximization

maximize_options: `difficult`, `technique(`*algorithm_spec*`)`, `iterate(#)`, `[no]log`, `trace`, `gradient`, `showstep`, `hessian`, `showtolerance`, `tolerance(#)`, `ltolerance(#)`, `nrtolerance(#)`, `nonrtolerance`, and `from(`*init_specs*`)`; see [R] **maximize**. These options are seldom used.

`technique(bhhh)` is not allowed.

The initial estimates must be specified as `from(`*matname* [`, copy`]`)`, where *matname* is the matrix containing the initial estimates and the `copy` option specifies that only the position of each element in *matname* is relevant. If `copy` is not specified, the column stripe of *matname* identifies the estimates.

The following option is available with `asclogit` but is not shown in the dialog box:

`coeflegend`; see [R] **estimation options**.

Remarks and examples

stata.com

`asclogit` fits McFadden's choice model (McFadden [1974]; for a brief introduction, see Greene [2012, sec. 18.2] or Cameron and Trivedi [2010, sec. 15.5]). In this model, we have a set of unordered alternatives indexed by $1, 2, \ldots, J$. Let y_{ij}, $j = 1, \ldots, J$, be an indicator variable for the alternative actually chosen by the ith individual (case). That is, $y_{ij} = 1$ if individual i chose alternative j and $y_{ij} = 0$ otherwise. The independent variables come in two forms: alternative specific and case specific. Alternative-specific variables vary among the alternatives (as well as cases), and case-specific variables vary only among cases. Assume that we have p alternative-specific variables so that for case i we have a $J \times p$ matrix, $\mathbf{X}_i$. Further, assume that we have q case-specific variables so that we have a $1 \times q$ vector $\mathbf{z}_i$ for case i. Our random-utility model can then be expressed as

$$\mathbf{u}_i = \mathbf{X}_i\boldsymbol{\beta} + (\mathbf{z}_i\mathbf{A})' + \boldsymbol{\epsilon}_i$$

Here $\boldsymbol{\beta}$ is a $p \times 1$ vector of alternative-specific regression coefficients and $\mathbf{A} = (\boldsymbol{\alpha}_1, \ldots, \boldsymbol{\alpha}_J)$ is a $q \times J$ matrix of case-specific regression coefficients. The elements of the $J \times 1$ vector $\boldsymbol{\epsilon}_i$ are independent Type I (Gumbel-type) extreme-value random variables with mean γ (the Euler–Mascheroni constant, approximately 0.577) and variance $\pi^2/6$. We must fix one of the $\boldsymbol{\alpha}_j$ to the constant vector to normalize the location. We set $\boldsymbol{\alpha}_k = 0$, where k is specified by the `basealternative()` option. The vector $\mathbf{u}_i$ quantifies the utility that the individual gains from the J alternatives. The alternative chosen by individual i is the one that maximizes utility.

McFadden's choice model is a specific case of conditional logistic regression. See [R] **clogit** for a more general application of conditional logistic regression. For example, `clogit` would be used when you have grouped data where each observation in a group may be a different individual, but all individuals in a group have a common characteristic. You may use `clogit` to obtain the same estimates as `asclogit` by specifying the `case()` variable as the `group()` variable in `clogit` and generating variables that interact the `casevars()` in `asclogit` with each alternative (in the form of an indicator variable), excluding the interaction variable associated with the base alternative. `asclogit` takes care of this data management burden for you.

▷ Example 1

We have data on 295 consumers and their choice of automobile. Each consumer chose among an American, Japanese, or European car; the variable `car` indicates the nationality of the car for each alternative. We want to explore the relationship between the choice of `car` to the consumer's sex (variable `sex`) and income (variable `income` in thousands of dollars). We also have information on the number of dealerships of each nationality in the consumer's city in the variable `dealer` that we want to include as a regressor. We assume that consumers' preferences are influenced by the number of dealerships in an area but that the number of dealerships is not influenced by consumer preferences (which we admit is a rather strong assumption). The variable `dealer` is an alternative-specific variable ($\mathbf{X}_i$ is a 3×1 vector in our previous notation), and `sex` and `income` are case-specific variables ($\mathbf{z}_i$ is a 1×2 vector). Each consumer's chosen car is indicated by the variable `choice`.

Let's list some of the data.

```
. use http://www.stata-press.com/data/r14/choice
. list id car choice dealer sex income in 1/12, sepby(id)
```

	id	car	choice	dealer	sex	income
1.	1	American	0	18	male	46.7
2.	1	Japan	0	8	male	46.7
3.	1	Europe	1	5	male	46.7
4.	2	American	1	17	male	26.1
5.	2	Japan	0	6	male	26.1
6.	2	Europe	0	2	male	26.1
7.	3	American	1	12	male	32.7
8.	3	Japan	0	6	male	32.7
9.	3	Europe	0	2	male	32.7
10.	4	American	0	18	female	49.2
11.	4	Japan	1	7	female	49.2
12.	4	Europe	0	4	female	49.2

We see, for example, that the first consumer, a male earning $46,700 per year, chose to purchase a European car even though there are more American and Japanese car dealers in his area. The fourth consumer, a female earning $49,200 per year, purchased a Japanese car.

We now fit our model.

```
. asclogit choice dealer, case(id) alternatives(car) casevars(sex income)
Iteration 0:   log likelihood = -273.55685
Iteration 1:   log likelihood = -252.75109
Iteration 2:   log likelihood = -250.78555
Iteration 3:   log likelihood =  -250.7794
Iteration 4:   log likelihood =  -250.7794
Alternative-specific conditional logit         Number of obs      =        885
Case variable: id                              Number of cases    =        295
Alternative variable: car                      Alts per case: min =          3
                                                              avg =        3.0
                                                              max =          3
                                                  Wald chi2(5)    =      15.86
Log likelihood =  -250.7794                       Prob > chi2     =     0.0072

------------------------------------------------------------------------------
      choice |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
car          |
      dealer |   .0680938   .0344465     1.98   0.048       .00058    .1356076
-------------+----------------------------------------------------------------
American     |  (base alternative)
-------------+----------------------------------------------------------------
Japan        |
         sex |  -.5346039   .3141564    -1.70   0.089    -1.150339    .0811314
      income |   .0325318    .012824     2.54   0.011     .0073973    .0576663
       _cons |  -1.352189   .6911829    -1.96   0.050    -2.706882    .0025049
-------------+----------------------------------------------------------------
Europe       |
         sex |   .5704109   .4540247     1.26   0.209    -.3194612    1.460283
      income |    .032042   .0138676     2.31   0.021      .004862    .0592219
       _cons |  -2.355249   .8526681    -2.76   0.006    -4.026448   -.6840501
------------------------------------------------------------------------------
```

Displaying the results as odds ratios makes interpretation easier.

```
. asclogit, or noheader
```

choice	Odds Ratio	Std. Err.	z	P>|z|	[95% Conf.	Interval]
car						
dealer	1.070466	.0368737	1.98	0.048	1.00058	1.145232
American	(base alternative)					
Japan						
sex	.5859013	.1840647	-1.70	0.089	.3165294	1.084513
income	1.033067	.013248	2.54	0.011	1.007425	1.059361
_cons	.2586735	.1787907	-1.96	0.050	.0667446	1.002508
Europe						
sex	1.768994	.8031669	1.26	0.209	.7265404	4.307178
income	1.032561	.0143191	2.31	0.021	1.004874	1.061011
_cons	.0948699	.0808925	-2.76	0.006	.0178376	.5045693

These results indicate that men (sex = 1) are less likely to pick a Japanese car over an American car than women (odds ratio 0.59) but that men are more likely to choose a European car over an American car (odds ratio 1.77). Raising a person's income increases the likelihood that he or she purchases a Japanese or European car; interestingly, the effect of higher income is about the same for these two types of cars.

◁

❑ Technical note

McFadden's choice model is related to multinomial logistic regression (see [R] **mlogit**). If all the independent variables are case specific, then the two models are identical. We verify this supposition by running the previous example without the alternative-specific variable, `dealer`.

```
. asclogit choice, case(id) alternatives(car) casevars(sex income) nolog
Alternative-specific conditional logit         Number of obs      =        885
Case variable: id                              Number of cases    =        295

Alternative variable: car                      Alts per case: min =          3
                                                              avg =        3.0
                                                              max =          3

                                                  Wald chi2(4)    =      12.53
Log likelihood = -252.72012                       Prob > chi2     =     0.0138
```

choice	Coef.	Std. Err.	z	P>|z|	[95% Conf.	Interval]
American	(base alternative)					
Japan						
sex	-.4694799	.3114939	-1.51	0.132	-1.079997	.141037
income	.0276854	.0123666	2.24	0.025	.0034472	.0519236
_cons	-1.962652	.6216804	-3.16	0.002	-3.181123	-.7441807
Europe						
sex	.5388441	.4525279	1.19	0.234	-.3480942	1.425782
income	.0273669	.013787	1.98	0.047	.000345	.0543889
_cons	-3.180029	.7546837	-4.21	0.000	-4.659182	-1.700876

To run `mlogit`, we must rearrange the dataset. `mlogit` requires a dependent variable that indicates the choice—1, 2, or 3—for each individual. We will use `car` as our dependent variable for those observations that represent the choice actually chosen.

```
. keep if choice == 1
(590 observations deleted)

. mlogit car sex income

Iteration 0:   log likelihood =  -259.1712
Iteration 1:   log likelihood = -252.81165
Iteration 2:   log likelihood = -252.72014
Iteration 3:   log likelihood = -252.72012

Multinomial logistic regression                 Number of obs     =        295
                                                LR chi2(4)        =      12.90
                                                Prob > chi2       =     0.0118
Log likelihood = -252.72012                     Pseudo R2         =     0.0249

------------------------------------------------------------------------------
         car |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
American     |  (base outcome)
-------------+----------------------------------------------------------------
Japan        |
         sex |  -.4694798   .3114939    -1.51   0.132    -1.079997    .1410371
      income |   .0276854   .0123666     2.24   0.025     .0034472    .0519236
       _cons |  -1.962651   .6216803    -3.16   0.002    -3.181122   -.7441801
-------------+----------------------------------------------------------------
Europe       |
         sex |   .5388443   .4525278     1.19   0.234     -.348094    1.425783
      income |    .027367    .013787     1.98   0.047      .000345    .0543889
       _cons |   -3.18003   .7546837    -4.21   0.000    -4.659182   -1.700877
------------------------------------------------------------------------------
```

The results are the same except for the model statistic: `asclogit` uses a Wald test and `mlogit` uses a likelihood-ratio test. If you prefer the likelihood-ratio test, you can fit the constant-only model for `asclogit` followed by the full model and use [R] **lrtest**. The following example will carry this out.

```
. use http://www.stata-press.com/data/r14/choice, clear

. asclogit choice, case(id) alternatives(car)

. estimates store null

. asclogit choice, case(id) alternatives(car) casevars(sex income)

. lrtest null .
```

❑

❑ Technical note

We force you to explicitly identify the case-specific variables in the `casevars()` option to ensure that the program behaves as you expect. For example, an `if` or `in` qualifier may drop observations in such a way that (what was expected to be) an alternative-specific variable turns into a case-specific variable. Here you would probably want `asclogit` to terminate instead of interacting the variable with the alternative indicators. This situation could also occur if `asclogit` drops cases, or observations if you use the `altwise` option, because of missing values.

❑

Stored results

asclogit stores the following in e():

Scalars

e(N)	number of observations
e(N_case)	number of cases
e(k)	number of parameters
e(k_alt)	number of alternatives
e(k_indvars)	number of alternative-specific variables
e(k_casevars)	number of case-specific variables
e(k_eq)	number of equations in e(b)
e(k_eq_model)	number of equations in overall model test
e(df_m)	model degrees of freedom
e(ll)	log likelihood
e(N_clust)	number of clusters
e(const)	constant indicator
e(i_base)	base alternative index
e(chi2)	χ^2
e(F)	F statistic
e(p)	significance
e(alt_min)	minimum number of alternatives
e(alt_avg)	average number of alternatives
e(alt_max)	maximum number of alternatives
e(rank)	rank of e(V)
e(ic)	number of iterations
e(rc)	return code
e(converged)	1 if converged, 0 otherwise

Macros

e(cmd)	asclogit
e(cmdline)	command as typed
e(depvar)	name of dependent variable
e(indvars)	alternative-specific independent variable
e(casevars)	case-specific variables
e(case)	variable defining cases
e(altvar)	variable defining alternatives
e(alteqs)	alternative equation names
e(alt#)	alternative labels
e(wtype)	weight type
e(wexp)	weight expression
e(title)	title in estimation output
e(clustvar)	name of cluster variable
e(offset)	linear offset variable
e(chi2type)	Wald, type of model χ^2 test
e(vce)	*vcetype* specified in vce()
e(vcetype)	title used to label Std. Err.
e(opt)	type of optimization
e(which)	max or min; whether optimizer is to perform maximization or minimization
e(ml_method)	type of ml method
e(user)	name of likelihood-evaluator program
e(technique)	maximization technique
e(datasignature)	the checksum
e(datasignaturevars)	variables used in calculation of checksum
e(properties)	b V
e(estat_cmd)	program used to implement estat
e(predict)	program used to implement predict
e(marginsnotok)	predictions disallowed by margins
e(asbalanced)	factor variables fvset as asbalanced
e(asobserved)	factor variables fvset as asobserved

Matrices

`e(b)`	coefficient vector
`e(stats)`	alternative statistics
`e(altvals)`	alternative values
`e(altfreq)`	alternative frequencies
`e(alt_casevars)`	indicators for estimated case-specific coefficients—`e(k_alt)`×`e(k_casevars)`
`e(ilog)`	iteration log (up to 20 iterations)
`e(gradient)`	gradient vector
`e(V)`	variance–covariance matrix of the estimators
`e(V_modelbased)`	model-based variance

Functions

`e(sample)`	marks estimation sample

Methods and formulas

In this model, we have a set of unordered alternatives indexed by $1, 2, \dots, J$. Let y_{ij}, $j = 1, \dots, J$, be an indicator variable for the alternative actually chosen by the ith individual (case). That is, $y_{ij} = 1$ if individual i chose alternative j and $y_{ij} = 0$ otherwise. The independent variables come in two forms: alternative specific and case specific. Alternative-specific variables vary among the alternatives (as well as cases), and case-specific variables vary only among cases. Assume that we have p alternative-specific variables so that for case i we have a $J \times p$ matrix, $\mathbf{X}_i$. Further, assume that we have q case-specific variables so that we have a $1 \times q$ vector $\mathbf{z}_i$ for case i. The deterministic component of the random-utility model can then be expressed as

$$
\begin{aligned}
\boldsymbol{\eta}_i &= \mathbf{X}_i\boldsymbol{\beta} + (\mathbf{z}_i\mathbf{A})' \\
&= \mathbf{X}_i\boldsymbol{\beta} + (\mathbf{z}_i \otimes \mathbf{I}_J)\,\mathrm{vec}(\mathbf{A}') \\
&= (\mathbf{X}_i,\ \mathbf{z}_i \otimes \mathbf{I}_J) \begin{pmatrix} \boldsymbol{\beta} \\ \mathrm{vec}(\mathbf{A}') \end{pmatrix} \\
&= \mathbf{X}_i^* \boldsymbol{\beta}^*
\end{aligned}
$$

As before, $\boldsymbol{\beta}$ is a $p \times 1$ vector of alternative-specific regression coefficients, and $\mathbf{A} = (\boldsymbol{\alpha}_1, \dots, \boldsymbol{\alpha}_J)$ is a $q \times J$ matrix of case-specific regression coefficients; remember that we must fix one of the $\boldsymbol{\alpha}_j$ to the constant vector to normalize the location. Here $\mathbf{I}_J$ is the $J \times J$ identity matrix, vec() is the vector function that creates a vector from a matrix by placing each column of the matrix on top of the other (see [M-5] **vec( )**), and $\otimes$ is the Kronecker product (see [M-2] **op_kronecker**).

We have rewritten the linear equation so that it is a form that can be used by `clogit`, namely, $\mathbf{X}_i^*\boldsymbol{\beta}^*$, where

$$
\mathbf{X}_i^* = (\mathbf{X}_i,\ \mathbf{z}_i \otimes \mathbf{I}_J)
$$

$$
\boldsymbol{\beta}^* = \begin{pmatrix} \boldsymbol{\beta} \\ \mathrm{vec}(\mathbf{A}') \end{pmatrix}
$$

With this in mind, see *Methods and formulas* in [R] **clogit** for the computational details of the conditional logit model.

This command supports the clustered version of the Huber/White/sandwich estimator of the variance using `vce(robust)` and `vce(cluster` *clustvar*`)`. See [P] **_robust**, particularly *Maximum likelihood estimators* and *Methods and formulas*. Specifying `vce(robust)` is equivalent to specifying `vce(cluster` *casevar*`)`, where *casevar* is the variable that identifies the cases.

Daniel Little McFadden was born in 1937 in North Carolina. He studied physics, psychology, and economics at the University of Minnesota and has taught economics at Pittsburgh, Berkeley, MIT, and the University of Southern California. His contributions to logit models were triggered by a student's project on freeway routing decisions, and his work consistently links economic theory and applied problems. In 2000, he shared the Nobel Prize in Economics with James J. Heckman.

References

Cameron, A. C., and P. K. Trivedi. 2010. *Microeconometrics Using Stata*. Rev. ed. College Station, TX: Stata Press.

Greene, W. H. 2012. *Econometric Analysis*. 7th ed. Upper Saddle River, NJ: Prentice Hall.

McFadden, D. L. 1974. Conditional logit analysis of qualitative choice behavior. In *Frontiers in Econometrics*, ed. P. Zarembka, 105–142. New York: Academic Press.

Also see

[R] **asclogit postestimation** — Postestimation tools for asclogit

[R] **asmprobit** — Alternative-specific multinomial probit regression

[R] **asroprobit** — Alternative-specific rank-ordered probit regression

[R] **clogit** — Conditional (fixed-effects) logistic regression

[R] **logistic** — Logistic regression, reporting odds ratios

[R] **logit** — Logistic regression, reporting coefficients

[R] **nlogit** — Nested logit regression

[R] **ologit** — Ordered logistic regression

[U] **20 Estimation and postestimation commands**